

Tutorial 1

Week of January 15, 2007

1 CFGs

Give CFG grammars (using BNF notation) for the following languages or say it cannot be done

1. All strings over the alphabet $\{0,1,2,3\}$ sorted in non-increasing order.

Answer:

```
<S> ::= <threes> <twos> <ones> <zeros> | epsilon
<threes> ::= 3 <threes> | epsilon
<twos> ::= 2 <twos> | epsilon
<ones> ::= 1 <ones> | epsilon
<zeros> ::= 0 <zeros> | epsilon
```

2. All strings over the alphabet $\{a,b\}$, in which every b is both immediately preceded by and followed by at least one a.

Answer:

```
<S> ::= a <BA> <S> | epsilon
<BA> ::= b a <BA> | epsilon
```

2 ambiguity

- a) $\langle S \rangle ::= \langle S \rangle \text{ and } \langle S \rangle \mid \langle S \rangle \text{ or } \langle S \rangle \mid a \mid b \mid c$

Is this grammar ambiguous?

Yes. Two distinct parse trees for string ‘‘a and b or c’’.

Develop a grammar that generates the same strings, but is not ambiguous.

```
<S> ::= <OR>
<OR> ::= <AND> | <OR> or <AND>
<AND> ::= <VAR> | <AND> and <VAR>
<VAR> ::= a | b | c
```

3 English descriptions

Give English descriptions of the languages described by each of the following grammars.

a) $\langle S \rangle ::= \langle S \rangle (\langle S \rangle) \langle S \rangle \mid \text{epsilon}$

Answer: The language of balanced parentheses.

b) $\langle S \rangle ::= \langle S \rangle a \langle S \rangle a \langle S \rangle \mid \langle S \rangle a \langle S \rangle b \langle S \rangle \mid$
 $\langle S \rangle b \langle S \rangle a \langle S \rangle \mid \langle S \rangle b \langle S \rangle b \langle S \rangle \mid \text{epsilon}$

Answer: All even length strings over $\{a,b\}$.